Light Intensity Experiments

Introduction

In dealing with experimental work done by students in virtual reality and 3D animation, we observed that worlds textured with student created pictures never looked quite as good as those textured with Microsoft supplied pictures. Figure 1, for example, is from the Microsoft DirectX SDK.

Figure 1   A picture from the Microsoft DirectX SDK.



The reason the Microsoft pictures look better is that lighting effects have been pre-applied. Even when displayed with a simplistic renderer (without lighting facilities), it still looks good. 



To better understand this effect, we thought it would be interesting to build an editor that permits lighting effects to be added to existing pictures. We started prototyping an elaborate interface that would deal with spot lights directed from arbitrary 3D locations but we gave that up when we realized we didn’t have the time to deal with such a complex application.



So we opted instead for a much simpler variant that is illustrated in Figure 2. This version permits a user to open a picture, select a rectangular area on the picture, and then either bump up or bump down the lighting intensity. Figure 3 illustrates the consequence. 

Figure 2   The LightApplicator application.

Figure 3   After selecting a rectangular area to effect.



Designing The Application

We built the application in Smalltalk expression using the supplied window builder. The application includes a file menu with two entries open and save, a graph pane for displaying the picture, an entry field to display the current intensity value (typically, a number between 0.0 and 1.0), and three buttons (one to select the area to be affected and the other two to increase or decrease the intensity; currently, in steps of 0.05).



To deal properly with lighting effects, we can’t make use of palettes. As a result, we only handle 24 bit pictures. Additionally, we don’t have any facilities to modify just one pixel in the 24-bit picture. What we need to do is extract all the pixels, modify the pixel or pixels we want to change, and then replace all the pixels. To deal with that we added two methods asRGBArrays and fromRGBArrays: (see Listing 1). The former extracts all the bits from a bitmap (assuming it is a 24-bit bitmap) and stores them in an array of size “bitmap height”. Each entry in turn is a subarray of size “bitmap width” containing instances of class Color. Class Color is a simple class already in the Smalltalk library which contains 3 instance variables and the corresponding methods to access them (red, red:, green, green:, blue, and blue:).



We also added a convenience method fromFileDialog which opens a file dialog and then reads in the bitmap file selected by the user.



The LightApplicator Class



The simple LightApplicator class (see Listing 2) was designed to keep track of the original bitmap (in “bitmap”) and a copy, the modified bitmap (in “modifiedBitmap”), along with the light intensity (in “ambientLighting”) and the boundary of the area to be modified (in “lightBoundary”). For convenience, we also cached three values “bitmapAsRGBArrays,” “modifiedBitmapAsRGBArrays,” and “drawingPane.” We will discuss the last variable “externalLink” later.



When file menu item “open” is selected, clickedFileOpen executes. It’s task is to prompt for a bitmap file which is recorded in “bitmap”. A copy is made and stored in “modifiedBitmap”. The corresponding RGB arrays are also obtained. In case a earlier selection on the previous bitmap had been made, the light boundary is eliminated (by setting it to nil). An update is them executed which triggers the #getContents events of all panes. In our case, we have handlers for the graph pane (see updatePicture: which forces the modified bitmap to be displayed) and the entry field (see updateAmbientLightingPane: which displays the current value of  “ambientLighting”).



After reading in a new bitmap, a user normally clicks on the “Select Area” button which causes the #clicked event handler clickedSelectArea to execute. “Display rectangleFromUser” prompts the user for a rectangular area which is returned as a rectangle in global screen coordinates (0@0 is the top-left corner of the screen). Method “mapToWindow:” converts it to pane coordinates (0@0 is the top-left corner of the pane). Because the bitmap is stretched (or shrunk) to exactly fill the pane, we need to map this rectangle called “area” to the corresponding parts of the bitmap. To do this, we compute a relative rectangle in “relativeArea”; i.e., a rectangle whose origin and extent are in the range 0.0 to 1.0. If we multiply these rectangle components by the bitmap’s extent, we get values that are in the correct range for the bitmap.



Once the light boundary has been calculated, we apply the lighting effect to the existing bitmap via method applyLighting (to be discussed shortly) and then we update the application. Similarly, by clicking on the “Increase Lighting” and “Decrease Lighting” buttons, we increment/decrement “ambientLighting”, apply the lighting effect, and update.



Applying the lighting effect (method applyLighting) amounts to recomputing the modified RGB arrays from the existing normal bitmap RGB arrays, and using it to construct a new modified bitmap (after releasing the previous one). Each color component is computed similarly; the red component, for example, is computed as follows:



	newColor red: (oldColor red * intensity)



where intensity is the ambient lighting value (normally, a value between 0.0 and 1.0 although we do permit the intensity to reach 2.0). Each color component must be an integer between 0 and 255. So we truncate it (it’s initially a float) and then ensure it is between the 0..255 limit as follows:



	newColor red: (((oldColor red * intensity) min: 255) max: 0)



Note that no modification is made to the color if the pixel coordinate x@y is outside the bounds of the light boundary rectangle.



Running in 16-bit mode



To test our application, we made sure we were not running in 8-bit mode. Our test compute didn’t support 24-bit color but it did support 16-bit color. It was interesting to watch the change as the intensity value varied between 0.0 and 1.0. However, we weren’t sure that the graph pane was actually displaying RGB values in 16-bit color. If it was actually using a palette, we would only see the nearest matching color in the color palette.



To verify that palettes were not being used, we decided to simultaneously display the modified picture using a 3D renderer that we had previous developed in a Computer Graphics course. We added one more variable “externalLink” to contain 3 values needed by the renderer and also added method tickDynamicLink which was invoked at the end of applyLighting. The method is designed to draw 3 cubes, one of which is textured with our modified bitmap. Figure 4 shows the resulting rendered cube. 



�



Figure 4   The modified bitmap textured over a cube.



The code, incorporating methods tickDynamicLink, modifiedBitmapAsPicture, setupDynamicLink, wrapupDynamicLink, and showDynamicLink, is shown categorized as “external rendering” for information only. It’s not clear that any of it will make sense on it’s own because it part of a very large package. If you do have a look, you need to know that groupDo: is a special sequencing method that permits code such as the following to execute.



	#(10 20 30 40) groupDo: [:v1 :v2 :v3 :v4  |

		… use v1, v2, v3, v4 …]



It is implemented very simple as follows in class Collection:



	groupDo: aBlock

    	aBlock evaluateWithArguments: self asArray



Conclusions

We implemented a simple application for performing light intensity experiments. It’s not a professional application by any means but it’s a start.

Listing 1 The Bitmap extensions.



class:								SpeedReadingWindow

superclass:						TextWindow 

instance variables:		speedMouse 



class methods

reading

fromFileDialog

    "Bitmap fromFileDialog"

    | fileName bitmap |

    fileName := FileDialog new

        title: 'Bitmap file...';

        shareAware;

        addFilter: '*.*' description: 'All Files (*.*)';

        addFilter: '*.bmp' description: 'Bitmap Files (*.bmp)';

        fileSpec: '*.bmp';

        open;

        file.

    fileName isNil ifTrue: [^nil "cancelled"].

    CursorManager execute changeFor: [

        bitmap := Bitmap fromFile: fileName].

    (bitmap isKindOf: Bitmap) ifTrue: [^bitmap].

    (MessageBox 

		confirm: 'This is not a bitmap file (*.bmp). Try again?')

			ifTrue: [^self fromFileDialog].

    ^nil



converting

fromRGBArrays: arrays

	| height width bits row base index color bitmap |

	height := arrays size. width := arrays first size.

	bits := ByteArray new: width * height * 3.

	1 to: height do: [:y |

		row := arrays at: y.

		base := (y - 1) * (3 * width).

		1 to: width do: [:x |

			color := row at: x.

			index := base + ((x - 1) * 3).

			bits at: index + 3 put: color red.

			bits at: index + 2 put: color green.

			bits at: index + 1 put: color blue]].

	bitmap := Bitmap new.

	bitmap bitmapInfo: (WinBitmapInfo 

		width: width height: height  planes: 1 bitCount: 24).

	bitmap createBitmap: bits.

	bitmap

		deviceContext: Bitmap memoryContext;

		graphicsTool: (Pen forDC: Bitmap memoryContext 

			medium: bitmap).

	^bitmap



instance methods



converting

asRGBArrays

	"An array of arrays; rows (one per y) of columns (one per x); each column is an instance of Color"

	| bits base index red green blue |

    self bitCount = 24 ifFalse: [self error: 'Only 24-bit bitmaps handled'].

	bits := self getDIBits.

	^(1 to: self height) collect: [:y |

		base := (y - 1) * (3 * self width).

		(1 to: self width) collect: [:x |

			index := base + ((x - 1) * 3).

			red := bits at: index + 3.

			green := bits at: index + 2.

			blue := bits at: index + 1.

			Color red: red green: green blue: blue]]





Listing 2 The LightApplicator class.



class:								LightApplicator

superclass:						ViewManager 

instance variables:		bitmap bitmapAsRGBArrays 

										modifiedBitmap 													modifiedBitmapAsRGBArrays 

										drawingPane ambientLighting 

										lightBoundary externalLink

pool dictionaries: 		ColorConstants 

										OperatingSystemConstants 



class methods

examples

example1

	"LightApplicator example1"

	self new open



instance methods

generated by window builder

createMenus: aPane

	… code not shown …

createViews

	… code not shown …



lighting effects

ambientLighting

	ambientLighting isNil ifTrue: [ambientLighting := 0.5].

	^ambientLighting

increaseAmbientLighting

	ambientLighting := (self ambientLighting + 0.05) min: 2.0

decreaseAmbientLighting

	ambientLighting := (self ambientLighting - 0.05) max: 0.0

 

opening/closing

finalize

	bitmap release.

	modifiedBitmap release.

	self wrapupDynamicLink.



file event handling

clickedFileOpen

	| newBitmap |

	newBitmap := Bitmap fromFileDialog.

	newBitmap isNil ifTrue: [^self]. "user cancelled"

	bitmap := newBitmap.

	bitmapAsRGBArrays := bitmap asRGBArrays.

	modifiedBitmap := bitmap clone.

	modifiedBitmapAsRGBArrays := 

		bitmapAsRGBArrays collect: [:array |

			array collect: [:color | color shallowCopy]].

	lightBoundary := nil.

	self update



clickedFileSave

	| fileName |

	fileName := (FileDialog new saveFile: '*.bmp') file.

	fileName isNil ifTrue: [^self "user cancelled"].

	modifiedBitmap outputToFile: fileName



light event handling

clickedSelectArea

	| area relativeArea |

	area := Display rectangleFromUser 

		mapToWindow: drawingPane.

	relativeArea := area relativeTo: drawingPane boundingBox.

	lightBoundary := (Rectangle

		origin: relativeArea origin * bitmap extent

		extent: relativeArea extent * bitmap extent) truncated.

	self applyLighting; update

   

clickedIncreaseLighting

	self increaseAmbientLighting; applyLighting; update



clickedDecreaseLighting

	self decreaseAmbientLighting; applyLighting; update



updating

update

	self mainView children do: [:child | child update]

updatePicture: aPane

	modifiedBitmap isNil ifTrue: [^self].

	aPane stretch: 1; bitmap: modifiedBitmap; showBitmap.

updateAmbientLightingPane: aPane

	aPane contents: self ambientLighting printString



applying lighting

applyLighting

	| border color intensity |

	bitmap isNil ifTrue: [^self].

	lightBoundary isNil ifTrue: [^self].

	intensity := self ambientLighting.

	0 to: bitmap width - 1 do: [:x |

		0 to: bitmap height - 1 do: [:y |

			color := ((bitmapAsRGBArrays at: y + 1) 

				at: x + 1) shallowCopy.

			(lightBoundary containsPoint: x@y) ifTrue: [

				color red: (((color red * intensity) truncated 

					min: 255) max: 0).

				color green: (((color green * intensity) truncated 

					min: 255) max: 0).

				color blue: (((color blue * intensity) truncated 

					min: 255) max: 0)].

			(modifiedBitmapAsRGBArrays at: y + 1) 

				at: x + 1 put: color]].

	modifiedBitmap release.

	modifiedBitmap :=  

		Bitmap fromRGBArrays: modifiedBitmapAsRGBArrays.

	self tickDynamicLink



external rendering

tickDynamicLink

	| texture |

	externalLink isNil ifTrue: [self setupDynamicLink].

	texture := Texture new 

		picture: self modifiedBitmapAsPicture.

	externalLink 

		groupDo: [:camera :cubeFrames :transformations |

			cubeFrames last facesDo: [:face |

				face texture clearPicture. face texture: texture]].

	self showDynamicLink

   

modifiedBitmapAsPicture

	| picturePointer color |

	bitmap isNil ifTrue: [^self].

	picturePointer := CanvasLibrary 

		newPictureWidth: bitmap width height: bitmap height.

	CanvasLibrary lockPicture: picturePointer.

	0 to: bitmap width - 1 do: [:x |

		0 to: bitmap height - 1 do: [:y |

			color := (modifiedBitmapAsRGBArrays at: y + 1) 

				at: x + 1.

			CanvasLibrary

				computePicture: picturePointer addressAt: x and: y;

				setPicture: picturePointer color: color red 

					and: color green and: color blue]].

	CanvasLibrary unlockPicture: picturePointer.

	^Picture add: (Picture new picturePointer: picturePointer)



setupDynamicLink

	| camera cubeFrames transformations extent |

	CanvasDLL setupPane: Transcript mainView handle.

	CanvasLibrary setupRenderer. Float clearExceptionFlags.

	camera := Camera new renderingMode: #textureRendering.

	camera textureRenderer: SpanRenderer new.

	extent := CanvasLibrary getCanvasWidth

		@CanvasLibrary getCanvasHeight.

	camera 

		changedScreenExtent: extent 

		depth: ((extent x + extent y)/4) negated 

		zScaling: 1.

	cubeFrames := Renderer example1Cubes.

	camera worldToCamera

		postTranslateBy: (Point3D x: 9.0 y: -3.0 z: 0.0).

	transformations := Array

		with: (Transformation4D 

			rotatedByDegrees: (Point3D x: 0.0 y: 10.0 z: 0.0))

		with: (Transformation4D 

			rotatedByDegrees: (Point3D x: 20.0 y: 0.0 z: 0.0))

		with: (Transformation4D 

			rotatedByDegrees: (Point3D x: 30.0 y: 10.0 z: 0.0)).

	externalLink := Array 

		with: camera with: cubeFrames with: transformations



wrapupDynamicLink

	externalLink isNil ifFalse: [

		externalLink 

			groupDo: [:camera :cubeFrames :transformations |

				cubeFrames do: [:frame |

					frame facesDo: [:face | 

						face texture picture finalize]]]].

	CanvasDLL wrapup

  

showDynamicLink

	externalLink 

		groupDo: [:camera :cubeFrames :transformations |

			camera tick; beginRendering.

			cubeFrames with: transformations do: [:frame :t |

				frame 

					localTransformation: t * 

						frame localTransformation; 

					invalidate.

				camera changedModel: frame transformation.

				frame facesDo: [:face | face render: camera]].

			camera beginLineRendering; endRendering.

			camera flip]

